;#######################################################################

;##

;## Contextual Query Language, Version 1.2

;##

;## This is a specification for the Contextual Query Language (CQL) [1]

;## grammar using Augmented Backus-Naur Form (ABNF) [2].

;##

;## Also referenced are the URI [3], IRI [4], XML 1.0 [5], XML 1.1 [6] 

;## specifications. Refer to [7] for the latest version of the

;## Unicode standard.

;##

;## Implementation Note: This ABNF makes use of generic Unicode

;## codepoints [7] for its charset which follow the XML 1.1 [6] spec.

;## It is intended that CQL users will be mindful of the XML 1.0 [5]

;## restrictions and will generate their queries accordingly. Note that

;## text literal strings are to be interpreted per the ABNF spec [2] as

;## being the US-ASCII subset of Unicode.

;## 

;## This ABNF is structured into Sections A-G, where Sections A-F form

;## a complete grammar for CQL at the token level. The additional section,

;## Section G, gives a mapping to Unicode codepoints.

;##

;## Edited: SRU Editorial Board [8]

;## Dated: 2008-02-01T16:00:00Z

;##

;## Copyright: (To be determined)

;##

;## References:

;##

;##   [1] http://www.loc.gov/standards/sru/specs/cql.html

;##   [2] http://www.ietf.org/rfc/rfc4234.txt

;##   [3] http://www.ietf.org/rfc/rfc3986.txt

;##   [4] http://www.ietf.org/rfc/rfc3987.txt

;##   [5] http://www.w3.org/TR/xml

;##   [6] http://www.w3.org/TR/xml11

;##   [7] http://www.unicode.org/unicode/standard/versions/

;##   [8] http://www.loc.gov/standards/sru/community/edboard.html

;##

;#######################################################################

;## A. CQL Query #######################################################

  cql-query          =  query [sort-spec]

  query              =  (prefix-assignment query) / scoped-clause                                      

  prefix-assignment  =  ">" [prefix "="] uri

  prefix             =  simple-name

  uri                =  quoted-identifier

  sort-spec          =  "sortby" 1*(index [modifier-list])

;## B. CQL Clauses #####################################################          

  scoped-clause      =  [scoped-clause boolean-group] search-clause

  search-clause      =  "(" query ")" / [index relation] search-term

  index              =  simple-name

  search-term        =  simple-string / quoted-string

;## C. CQL Booleans ####################################################

  boolean-group      =  boolean [modifier-list]

  boolean            =  "and" / "or" / "not" / "prox"

;## D. CQL Relations ###################################################

  relation           =  comparator [modifier-list]

  comparator         =  comparator-symbol / comparator-name

  comparator-symbol  =  "=" / ">" / "<" / ">=" / "<=" / "<>" / "=="

  comparator-name    =  simple-name

;## E. CQL Modifiers ###################################################

  modifier-list      =  1*modifier

  modifier           =  "/" modifier-name [modifier-relation]

  modifier-relation  =  comparator-symbol modifier-value

  modifier-name      =  simple-name

  modifier-value     =  simple-string / quoted-string

;## F. CQL Terminals ###################################################

  quoted-identifier  =  DQUOTE *UCHAR DQUOTE

    ; Double quotes enclosing a URI resource identifier. RFC 3986 [3]

    ; specifies the allowed characters which fall within the US-ASCII

    ; subset of Unicode.

    ;

    ; If IRI [4] support is required then UCHAR in above rule should be

    ; replaced by ICHAR and the rules for ICHAR, ICHAR-1, ICHAR-2 should

    ; be uncommented. 

  quoted-string      =  DQUOTE *CHAR DQUOTE

    ; Double quotes enclosing a sequence of any characters except double

    ; quote (unless preceded by backslash "\"). Backslash escapes the 

    ; character following it. The surrounding double quotes are not  

    ; included in the value.

  simple-name        =  *SCHAR

    ; Note that the following reserved words (case insensitive) are

    ; specifically excluded: "and", "or", "not", "prox", "sortby".

    ;

    ; In XML EBNF [5,6] this production would be written as:

    ; *SCHAR - ("and" / "or" / "not" / "prox" / "sortby")

  simple-string      =  *SCHAR

    ; See notes on reserved characters in <simple-name> rule above. 

;## G. Unicode Codepoints ##############################################

  CHAR               =  %x1-D7FF / %xE000-FFFD / %x10000-10FFFF

                        ; Any Unicode character, excluding the 

                        ; surrogate blocks, FFFE, and FFFF

  ; Note: CQL authors are encouraged to avoid "compatibility chara-

  ; cters", as defined in Unicode [Unicode]. The characters defined in

  ; the following ranges are also discouraged. They are either control

  ; characters or permanently undefined Unicode characters:

  ;

  ;     %x1-8, %xB-C, %xE-1F, %x7F-84, %x86-9F, %xFDD0-FDDF,

  ;     %x1FFFE-1FFFF, %x2FFFE-2FFFF, %x3FFFE-3FFFF,

  ;     %x4FFFE-4FFFF, %x5FFFE-5FFFF, %x6FFFE-6FFFF,

  ;     %x7FFFE-7FFFF, %x8FFFE-8FFFF, %x9FFFE-9FFFF,

  ;     %xAFFFE-AFFFF, %xBFFFE-BFFFF, %xCFFFE-CFFFF,

  ;     %xDFFFE-DFFFF, %xEFFFE-EFFFF, %xFFFFE-FFFFF,

  ;     %x10FFFE-10FFFF

  DQUOTE             =  %x22    ; double quote

  RCHAR              =  DQUOTE / "(" / ")" / "/" / "<" / "=" / ">"

                        ; Reserved character in <auoted-string>.

  SCHAR              =  CHAR 

                        ; Simple character excludes reserved characters

                        ; specified in the <RCHAR> rule above.

                        ;

                        ; In XML EBNF [5,6] this production would be

                        ; written as:

                        ; [^#x22#x28#x29#x2F#x3C#x3D#x3E]

  UCHAR              =  %x21 / %x23-3B / %x3D / %x3F-5B / %x5D / %x5F

                        / %x61-7A / %x7E

                        ; URI character. Compiled from RFC 3986 [3].

  ; Note: Uncomment these rules if IRI support required.

  ;

  ; ICHAR            =  UCHAR / ICHAR-1 / ICHAR-2

  ;

  ; ICHAR-1          =  %xA0-D7FF / %xF900-FDCF / %xFDF0-FFEF

  ;                     / %x10000-1FFFD / %x20000-2FFFD / %x30000-3FFFD

  ;                     / %x40000-4FFFD / %x50000-5FFFD / %x60000-6FFFD

  ;                     / %x70000-7FFFD / %x80000-8FFFD / %x90000-9FFFD

  ;                     / %xA0000-AFFFD / %xB0000-BFFFD / %xC0000-CFFFD

  ;                     / %xD0000-DFFFD / %xE1000-EFFFD

  ;                     ; Known as <ucschar> in RFC 3987 [4].

  ;

  ; ICHAR-2          =  %xE000-F8FF / %xF0000-FFFFD / %x100000-10FFFD

  ;                     ; Known as <iprivate> in RFC 3987 [4].

;#######################################################################

;## EOF

;#######################################################################
